

NAG Library Routine Document

G05LMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05LMF generates a vector of pseudorandom numbers from a two parameter Weibull distribution with shape parameter a and scale parameter b .

2 Specification

```
SUBROUTINE G05LMF(A, B, N, X, IGEN, ISEED, IFAIL)
INTEGER          N, IGEN, ISEED(4), IFAIL
double precision A, B, X(N)
```

3 Description

The distribution has PDF (probability density function)

$$f(x) = \frac{a}{b} x^{a-1} e^{-x^a/b} \quad \text{if } x > 0,$$

$$f(x) = 0 \quad \text{otherwise.}$$

G05LMF returns the value $(-b \ln y)^{1/a}$, where y is a pseudorandom number from a uniform distribution over $(0, 1)$.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05LMF.

4 References

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin
 Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

- | | | |
|----|--|---------------|
| 1: | A – double precision
<i>On entry:</i> a , the shape parameter of the distribution.
<i>Constraint:</i> $A > 0.0$. | <i>Input</i> |
| 2: | B – double precision
<i>On entry:</i> b , the scale parameter of the distribution.
<i>Constraint:</i> $B > 0.0$. | <i>Input</i> |
| 3: | N – INTEGER
<i>On entry:</i> n , the number of pseudorandom numbers to be generated.
<i>Constraint:</i> $N \geq 0$. | <i>Input</i> |
| 4: | X(N) – double precision array
<i>On exit:</i> the n pseudorandom numbers from the specified Weibull distribution. | <i>Output</i> |

- 5: IGEN – INTEGER *Input*
On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.
- 6: ISEED(4) – INTEGER array *Communication Array*
On entry: contains values which define the current state of the selected generator.
On exit: contains updated values defining the new state of the selected generator.
- 7: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $A \leq 0.0$.

IFAIL = 2

On entry, $B \leq 0.0$.

IFAIL = 3

On entry, $N < 0$.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints the first five pseudorandom real numbers from a Weibull distribution with shape parameter 1.0 and scale parameter 2.0, generated by a single call to G05LMF, after initialization by G05KBF.

9.1 Program Text

```

*      G05LMF Example Program Text
*      Mark 20 Release. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER          NOUT, N
      PARAMETER        (NOUT=6,N=5)
*      .. Local Scalars ..
      DOUBLE PRECISION A, B
      INTEGER          IFAIL, IGEN
*      .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          ISEED(4)
*      .. External Subroutines ..
      EXTERNAL         G05KBF, G05LMF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05LMF Example Program Results'
      WRITE (NOUT,*)
*      Initialize the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355
*      IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*
      A = 1.0D0
      B = 2.0D0
      IFAIL = 1
      CALL G05LMF(A,B,N,X,IGEN,ISEED,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
*
          WRITE (NOUT,99999) X
*
      ELSE
          WRITE (NOUT,99998) IFAIL
      END IF
*
99999 FORMAT (1X,F10.4)
99998 FORMAT (1X,' ** G05LMF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05LMF Example Program Results

```

4.8310
0.1006
1.8009
0.5936
0.1031

```
